

Genel: (004) Sort by Header

#PURPOSE: To sort a list view by the clicked-on header.

#Start, End Context: Any, Any

#2017-02-20, JH, Created script.

#-----INITIALIZE-----

Allow User Abort [Off]

Set Error Capture [On]

#-----POPULATE PARAMETERS, VARIABLES, CONSTANTS-----

Set Variable [\$scriptParameter; Value:If(not IsEmpty(Get(ScriptParameter)) ; Get(ScriptParameter) ; "{}")]

#Parameters

Set Variable [\$fieldObjectName; Value:JSONGetElement(\$scriptParameter ; "fieldObjectName")]

#Variables

Set Variable [\$\$windowSortStateList; Value:If(not IsEmpty(\$\$windowSortStateList) ; \$\$windowSortStateList ; "{}")]

Set Variable [\$currentWindowNameTO_Name; Value:Get(WindowName) & ":" & Get(LayoutTableName)]

#-----GRAB PREVIOUS SORT FIELD & ORDER-----

Set Variable [\$previousSortFieldName; Value:JSONGetElement(\$\$windowSortStateList ; \$currentWindowNameTO_Name & "." & "field")]

Set Variable [\$previousSortOrder; Value:JSONGetElement(\$\$windowSortStateList ; \$currentWindowNameTO_Name & "." & "sortOrder")]

#-----ENSURE THERE'S A FIELD TO SORT-----

If [IsEmpty (\$fieldObjectName)]

Exit Script [Result: 0]

End If

#-----GO TO FIELD-----

Go to Object [Object Name: \$fieldObjectName]

#-----SET CURRENT SORT FIELD AND DIRECTION-----

Set Variable [\$sortFieldName; Value:Get(ActiveFieldTableName) & ":" & Get (ActiveFieldName)]

Set Variable [\$sortOrder; Value:Case((\$previousSortFieldName = \$sortFieldName and \$previousSortOrder = "D") or \$previousSortFieldName ≠ \$sortFieldName ; "A" ; "D")]

#-----SORT-----

If [\$sortOrder = "A"]

#Ascending

Sort Records by Field [Ascending]

Else

#Descending

Sort Records by Field [Descending]

End If

#-----AFTER SORT GO TO FIRST RECORD-----

Go to Record/Request/Page

[First]

Scroll Window

[Home]

#-----ADD SORT STATE TO LIST-----

Set Variable [\$\$windowSortStateList; Value:JSONSetElement(\$\$windowSortStateList ;
[\$currentWindowNameTO_Name & "." & "field" ; \$sortFieldName ; JSONString] ;
[\$currentWindowNameTO_Name & "." & "sortOrder" ; \$sortOrder ; JSONString]
)]

// #-----WRAP UP-----

Exit Script [Result: 1]